

```
;- DL8MCG -- HAM-Radio-DTMF-Decoder mit Görtzel-Filter-Algorithmus --- auf UniDsp56-Board von DL9GFA --
```

```
dftDTMF_TH      equ    0.00003    ; Mindestamplitude
dftDTMF_N_CNT   equ    1000       ; Anzahl Mittelungen
constNumberOfFilters equ    8
```

```
org x:
```

```
AvgCnt          ds      1
```

```
DTMF_COS_697Hz  ds      1
DTMF_COS_770Hz  ds      1
DTMF_COS_852Hz  ds      1
DTMF_COS_941Hz  ds      1
DTMF_COS_1209Hz ds      1
DTMF_COS_1336Hz ds      1
DTMF_COS_1477Hz ds      1
DTMF_COS_1633Hz ds      1
```

```
DTMF_Q1_697Hz   ds      1
DTMF_Q1_770Hz   ds      1
DTMF_Q1_852Hz   ds      1
DTMF_Q1_941Hz   ds      1
DTMF_Q1_1209Hz  ds      1
DTMF_Q1_1336Hz  ds      1
DTMF_Q1_1477Hz  ds      1
DTMF_Q1_1633Hz  ds      1
```

```
DTMF_PWR_697Hz  ds      1
DTMF_PWR_770Hz  ds      1
DTMF_PWR_852Hz  ds      1
DTMF_PWR_941Hz  ds      1
DTMF_PWR_1209Hz ds      1
DTMF_PWR_1336Hz ds      1
DTMF_PWR_1477Hz ds      1
DTMF_PWR_1633Hz ds      1
```

```
org y:
```

```
DTMF_Q2_697Hz   ds      1
DTMF_Q2_770Hz   ds      1
DTMF_Q2_852Hz   ds      1
DTMF_Q2_941Hz   ds      1
DTMF_Q2_1209Hz  ds      1
DTMF_Q2_1336Hz  ds      1
DTMF_Q2_1477Hz  ds      1
DTMF_Q2_1633Hz  ds      1
```

```
org p:
```

```
constDTMF_COS_697Hz dc    @COS(2*PI*697/SAMPLE_FREQ)
constDTMF_COS_770Hz dc    @COS(2*PI*770/SAMPLE_FREQ)
constDTMF_COS_852Hz dc    @COS(2*PI*852/SAMPLE_FREQ)
constDTMF_COS_941Hz dc    @COS(2*PI*941/SAMPLE_FREQ)
constDTMF_COS_1209Hz dc   @COS(2*PI*1209/SAMPLE_FREQ)
constDTMF_COS_1336Hz dc   @COS(2*PI*1336/SAMPLE_FREQ)
constDTMF_COS_1477Hz dc   @COS(2*PI*1477/SAMPLE_FREQ)
constDTMF_COS_1633Hz dc   @COS(2*PI*1633/SAMPLE_FREQ)
```

```
;------ Init DTMF-Filter -----
```

```
DTMF_init      bclr    #F_DTMF_sent,x:Flags
               move    #constDTMF_COS_697Hz,r0
               move    #DTMF_COS_697Hz,r1
               do      #constNumberOfFilters,DTMF_init1
                   move    p:(r0)+,x0
                   move    x0,x:(r1)+
DTMF_init1     jmp     DTMF_init2
```

```
;------ Rx2-Tone-Filter-Bank (DTMF und Rufton) -----
```

```
DTMF_Filter    move    #>-1,m2
               move    #>-1,m3
               move    #>-1,m4
               move    #>DTMF_COS_697Hz,r2      ; cos(w), erstes Element in Ton-Tabelle
               move    #>DTMF_Q1_697Hz,r3       ; Q1, erstes Element in Ton-Tabelle
               move    #>DTMF_Q2_697Hz,r4       ; Q2, erstes Element in Ton-Tabelle
               move    #constNumberOfFilters,n0
```

```
;------ Anzahl Mittelungen erreicht? -----
```

```
               move    x:AvgCnt,a
               sub     #1,a
               jmi    DTMF_Power
               move    a,x:AvgCnt
```

```
;------ Audio-Wert holen und skalieren -----
```

```
               move    x:Tx2_AfDacVal,a
               asr     #8,a,a                    ; AF-Sample durch 2^8 teilen, x(i) -> a
               nop
               move    a,r0                      ; x(i) -> r0
```

```

;----- Görtzel-Filter -----
do      n0,DTMF_Goertzel
        move   x:(r3),x0 y:(r4),y0      ; Q1 -> x0 , Q2 -> y0
        sub    y0,a x:(r2)+,y0          ; x(i) - Q2 -> a , cos(w) -> y0
        mac    x0,y0,a x0,y:(r4)+      ; Q1 * cos(w) + x(i) - Q2 -> a , Q1 -> Q2
        mac    x0,y0,a                  ; 2 * Q1 * cos(w) + x(i) - Q2 -> a
        nop
        move   a,x:(r3)+                ; 2 * Q1 * cos(w) + x(i) - Q2 -> Q1
        move   r0,a
DTMF_Goertzel    rts
;----- Berechne Audio-Leistung -----
DTMF_Power
do      n0,DTMF_Power1
        move   x:(r3)+,x0 y:(r4)+,y0    ; Q1 -> x0 , Q2 -> y0
        mpy    x0,y0,b x:(r2)+,x1      ; Q1*Q2 -> b , cos(w) -> x1
        mpy    y0,y0,a                  ; Q2^2 -> a
        mac    x0,x0,a b,y1            ; Q2^2 + Q1^2 -> a , Q1*Q2 -> y1
        mac    -x1,y1,a                 ; Q2^2 + Q1^2 -cos(w)*Q1*Q2 -> a
        mac    -x1,y1,a                 ; Q2^2 + Q1^2 -2*cos(w)*Q1*Q2 -> a
        nop
        move   a,x:(r1)+
DTMF_Power1
;----- Bestimme stärksten Ton der Row-Frequency-Group -----
move   #dftDTMF_TH,x0
move   #>DTMF_PWR_941Hz,r0
jsr    DtmfMax
;----- Vergleich auf Mindestamplitude -----
jlt    DTMF_Power2
move   y0,b
asl    #2,b,b
nop
move   b,y1
;----- Bestimme stärksten Ton der Column-Frequency-Group -----
move   #dftDTMF_TH*80,x0
move   #>DTMF_PWR_1633Hz,r0
jsr    DtmfMax
;----- Vergleich auf Mindestamplitude -----
jlt    DTMF_Power2
move   y0,b
;----- Dekodiere Ton erkannten Ton -----
add    y1,b
;----- Ton in Dsp-Status-Wort eintragen und an uC2 senden -----
jset   #F_DTMF_sent,x:Flags,DTMF_init2
bset   #F_DTMF_sent,x:Flags
LedRtOn
move   x:DspStatusWord,a
and    #FFFFFF0,a
move   b,y1
or     y1,a
nop
move   a,x:DspStatusWord
bset   #DSW_DTMF_new,a
jsr    SendStatusDTMFToUc          ; DTMF-Zeichen an uC senden
jmp    DTMF_init2
;----- Keinen gültigen DTMF-Ton erkannt -----
DTMF_Power2    LedRtOff
bclr   #F_DTMF_sent,x:Flags
;----- Nächsten Auswerte-Zyklus initialisieren -----
DTMF_init2
move   #dftDTMF_N_CNT,x0
move   x0,x:AvgCnt
move   #>DTMF_Q1_697Hz,r3          ; Q1
move   #>DTMF_Q2_697Hz,r4          ; Q2
move   #constNumberOfFilters,n0   ; count
move   #0,x0
do     n0,DTMF_clr
        move   x0,x:(r3)+          ; Q1 = 0
        move   x0,y:(r4)+          ; Q2 = 0
DTMF_clr    rts
;----- Bestimmung des Gruppen-Tones mit der höchsten Amplitude und Prüfung auf Mindestamplitude -----
;* Input  : r0 = Index auf höchsten Ton der jeweiligen Gruppe
;          x0 = Schwellwert (Mindestamplitude)
; Output  : Negativ/Zero-Flags
DtmfMax
move   #3,b0
move   x:(r0)-,x1
move   b0,y0
do     b0,DtmfMax2
        dec    b
        move   x:(r0)-,a
        cmp    x1,a
        jle    DtmfMax1
        move   a,x1                ; neue Schwelle
        move   b0,y0
DtmfMax1    nop
DtmfMax2    move   x1,a
            cmp    x0,a
            rts
;-----

```